Ukážky použitia MATHEMATICE 6 na strednej škole

Mnohočleny (1.ročník gymnázia)

najčastejšie riešené úlohy boli:

- určenie hodnoty mnohočlena
- úprava mnohočlenov
- delenie mnohočlenov
- rozklad mnohočlenov na súčin
- racionálne lomené výrazy (určenie definičného oboru)
- výrazy s absolútnou hodnotou

Určenie hodnoty mnohočlena

```
A[t_] := 8 t^3 - 4 t^2 + 2 t + 1 / 2
```

```
A[3]
```

$$\frac{373}{2}$$

```
A[x / 2]
```

$$\frac{1}{2} + x - x^2 + x^3$$

```
A[3.5]
```

301.5

```
Table[A[t], {t, 0, 10}]
```

$$\left\{\frac{1}{2}, \frac{13}{2}, \frac{105}{2}, \frac{373}{2}, \frac{913}{2}, \frac{1821}{2}, \frac{3193}{2}, \frac{5125}{2}, \frac{7713}{2}, \frac{11\,053}{2}, \frac{15\,241}{2}\right\}$$

```
Table[A[t], {t, 0, 10}] // N
```

{0.5, 6.5, 52.5, 186.5, 456.5, 910.5, 1596.5, 2562.5, 3856.5, 5526.5, 7620.5}

```
Manipulate[A[x], {x, 0, 10, 1}]
```



Úprava mnohočlenov

Pohľad študenta-nepamätám si presne vzorec

```
(a + b)^2 // Expand
```

$a^2 + 2\,b\,a + b^2$

```
(a + b + c)^3 // Expand
```

$a^3 + 3\,b\,a^2 + 3\,c\,a^2 + 3\,b^2\,a + 3\,c^2\,a + 6\,b\,c\,a + b^3 + c^3 + 3\,b\,c^2 + 3\,b^2\,c$

Pohľad učiteľa

- chcem zostaviť príklad, ktorý bude mať ľahko kontrolovateľný výsledok
- chcem rýchlo skontrolovať výsledky písomky

```
(m^2 + m)^2 + (m - 1) (m^2 + 1) - m^4 // Simplify
```

$3\,m^3 + m - 1$

```
A[t] * A[t]
```

$\left(8\,t^3 - 4\,t^2 + 2\,t + \frac{1}{2}\right)^2$

```
A[t] * A[t] // Expand
```

$$64\, t^6 - 64\, t^5 + 48\, t^4 - 8\, t^3 + 2\, t + \frac{1}{4}$$

chcem poznať koeficient pri niektorej mocnine výrazu

```
Coefficient[ A[t]^2, t^4]
```

48

```
CoefficientList[ A[t]^2, t]
```

$$\left\{\frac{1}{4}, 2, 0, -8, 48, -64, 64\right\}$$

Delenie mnohočlenov

```
(9 x^2 - 5 x^2 + 5 x - 20) / (x - 4)
```

$$\frac{-20 + 5\, x + 4\, x^2}{-4 + x}$$

```
PolynomialQuotient[(9 x^2 - 5 x^2 + 5 x - 20), (x - 4), x]
```

$21 + 4\, x$

```
PolynomialRemainder[(9 x^2 - 5 x^2 + 5 x - 20), (x - 4), x]
```

64

```
(9 x^2 - 5 x^2 + 5 x - 20) / (x - 4) // FullSimplify
```

$$21 + \frac{64}{-4 + x} + 4\, x$$

Rozklad mnohočlenov na súčin

```
x^2 - 6 x + 9 // Simplify
```

$(x-3)^2$

```
2 a^2 + a - 1 // Factor
```

$(a+1)(2a-1)$

```
(3 a - 4) ^2 - 8 (3 a - 4) + 16 // Factor
```

$(3a-8)^2$

Výrazy obsahujúce goniometrické funkcie

všetky základné vzorce Mathematica pozná a môžeme ich používať

```
TrigExpand[Sin[2 x]]
```

2 Cos[x] Sin[x]

```
TrigExpand[Sin[x - y] ]
```

Cos[y] Sin[x] - Cos[x] Sin[y]

príkaz **TrigFactor** - sa snaží zapísať výsledok ako súčin

```
vyraz2 = TrigExpand[Cos[2 x] Sin[3 x]]
```

$$\frac{\mathrm{Sin}[x]}{2}+\frac{5}{2}\mathrm{Cos}[x]^4\,\mathrm{Sin}[x]-5\,\mathrm{Cos}[x]^2\,\mathrm{Sin}[x]^3+\frac{\mathrm{Sin}[x]^5}{2}$$

```
TrigFactor[vyraz2]
```

$$2\,(1+2\,\mathrm{Cos}[2\,x])\,\mathrm{Sin}\left[\frac{\pi}{4}-x\right]\mathrm{Sin}[x]\,\mathrm{Sin}\left[\frac{\pi}{4}+x\right]$$

```
TrigFactor[vyraz2] // Simplify
```

$\frac{1}{2}(1+2\,\mathrm{Cos}[2\,x])\,(-\mathrm{Sin}[x]+\mathrm{Sin}[3\,x])$

TrigReduce sa snaží zapísať výraz ako súčet (nepoužíva súčin ani mocninu)

```
TrigReduce[vyraz2]
```

$\frac{1}{2}(\mathrm{Sin}[x]+\mathrm{Sin}[5\,x])$

```
vyraz =.
```

```
PowerExpand[Sqrt[x y]]
```

$\sqrt{x}\,\sqrt{y}$

```
PowerExpand[Log[a^b]]
```

b Log[a]

Najväčší spoločný deliteľ, najmenší spoločný násobok

- PolynomialGCD["poly"$_1$, "poly"$_2$, …] dáva najväčišho spoločného deliteľa polynómov "poly"$_i$.
- PolynomialLCM["poly"$_1$, "poly"$_2$, …] dáva najmenší spoločný násobok "poly"$_i$.

```
PolynomialGCD[y^2 - y, y^3 + y^2, y^2 - 1]
```

1

```
PolynomialLCM[y^2 - y, y^3 + y^2, y^2 - 1]
```

$(-1+y)\,y^2\,(1+y)$

Racionálne lomené výrazy

```
vyraz = (3 x^2 - 3 x y) / (3 (x - y) ^2)
```

$$\frac{3x^2 - 3xy}{3(x-y)^2}$$

```
vyraz // Simplify
```

$$\frac{x}{x-y}$$

```
vyraz1 =
  (y / (x^2 - x y) + x / (y^2 - x y)) * ((x^2 y + x y^2) / (x^2 - y^2))
```

$$\frac{(x^2 y + x y^2)\left(\frac{y}{x^2-x y}+\frac{x}{-x y+y^2}\right)}{x^2-y^2}$$

```
% // Simplify
```

$$-\frac{x+y}{x-y}$$

```
vyraz1 // Expand
```

$$\frac{x^2 y^2}{(x^2-x y)(x^2-y^2)}+\frac{x y^3}{(x^2-x y)(x^2-y^2)}+\frac{x^3 y}{(x^2-y^2)(-x y+y^2)}+\frac{x^2 y^2}{(x^2-y^2)(-x y+y^2)}$$

```
vyraz1 // Denominator
```

$$x^2 - y^2$$

```
vyraz1 // Numerator
```

$$(x^2 y + x y^2)\left(\frac{y}{x^2-x y}+\frac{x}{-x y+y^2}\right)$$

Výrazy, rovnice s absolútnou hodnotou

```
Clear[vyraz];
vyraz = Abs[x - 2] - 4 + Abs[x - 4]
```

```
-4 + Abs[-4 + x] + Abs[-2 + x]
```

Chceme riešiť rovnicu s absolútnou hodnotou. Mathematica 5 dávala výsledok

```
Solve[vyraz == 0, x]
```

```
Solve::ifun :
 Inverse functions are being used by Solve, so some solutions may not
   be found; use Reduce for complete solution information. More…
```

```
{{x → 1}, {x → 5}}
```

Mathematica 6 dávala výsledok

```
Solve[vyraz == 0, x]
```

```
{{x → 1}, {x → 5}}
```

```
Plot[{Abs[x - 2] + Abs[x - 4], 4}, {x, -4, 6}, PlotStyle ->
  {Thickness[0.01], {Thickness[0.01], Dashing[{0.06}]}}]
```



```
Plot[vyraz, {x, -2, 10}, PlotRange → All]
```




```
Abs[x - 2] + Abs[x - 4] == 4 /. x -> 1
```

```
True
```

```
Simplify[Abs[x - 2] + Abs[x - 4], x ≥ 4]
```

```
2 (-3 + x)
```

```
Simplify[vyraz, 2 <= x ≤ 4]
```

```
-2
```

Mathematica 6 má prepracovanejší algoritmus na riešenie rovníc a absolútnou hodnotou.

Mathematica 5 dávala výsledok

```
Simplify[vyraz, x < 2]
```

$|x-4|+|x-2|-4$

Mathematica 6 dávala výsledok - má už zlepšené algoritmy.

```
Simplify[vyraz, x < 2]
```

```
2 - 2 x
```

Pomerne jednoducho môžeme vypočítať aj hodnotu výrazu s absolútnou hodnotou. (pomocou dosadzovacieho operátora)

```
vyraz /. x → -4
```

```
10
```

Chceme riešiť nerovnicu s absolútnymi hodnotami. *Mathematica* 5 potrebovala načítať špeciálny balík na riešenie nerovníc.

```
<< Algebra`InequalitySolve`
```

Mathematica 6 má už tento balík integrovaný.

```
InequalitySolve[{vyraz ≥ 5}, x]
```

$x \leq -\frac{3}{2} \ || \ x \geq \frac{15}{2}$

Môžeme sa pozrieť aj na grafickú prezentáciu výsledku nerovnice

```
Plot[{vyraz, 5}, {x, -6, 10}, Filling → 5]
```



Mathematica 5 používala na riešenie nerovníc príkaz Inequality Solve

```
InequalitySolve[{vyraz == 0, 2 <= x < 4}, x]
```

False

```
InequalitySolve[{vyraz == 0, x ≤ 2}, x]
```

$x == 1$

Mathematica 6 nahradila tent príkaz príkazom Reduce

```
Reduce[{vyraz == 0, 2 <= x < 4}, x]
```

```
False
```

```
Reduce[{vyraz == 0, x ≤ 2}, x]
```

$x == 1$

Ďalšie typy rovníc (1.ročník gymnázia)

najčastejšie riešené úlohy boli:

- rovnice s neznámou v menovateli
- riešenie nerovníc
- kvadratické rovnice (úprava dvojčlena na súčin, grafické riešenie, vzťahy medzi koeňmi a koeficientami)
- kvadratické nerovnice
- rovnice s parametrom

Rovnice s neznámou v menovateli

```
Clear[rovnica]
rovnica = (2 x + 1) / (x - 1)  + (x + 1) / (x - 1) == 11 / 2
```

$$\frac{1+x}{-1+x}+\frac{1+2x}{-1+x}==\frac{11}{2}$$

```
Simplify[rovnica]
```

$$\frac{-3+x}{-1+x}==0$$

Koreňom rovnice je číslo 3, patrí do definičného odoru.

```
Solve[rovnica, x]
```

$\{\{x \to 3\}\}$

```
Clear[rovnica1]
rovnica1 = 5 + 3 / (3 x - 12) == (5 - x) / (x - 4)
```

$$5+\frac{3}{-12+3\,x}==\frac{5-x}{-4+x}$$

```
Simplify[rovnica1]
```

```
False
```

Výpočtom dostaneme, ze koreňom by malo byť číslo 4, ktoré ale nepatrí do definičného oboru!

Výhoda: vďaka tejto vlastnosti nemusíme robiť skúšku správnosti

```
Solve[rovnica1]
```

```
{}
```

Riešenie nerovníc, nerovnice s neznámou v absolútnej hodnote

Mathematica 5 používala na výpočet príkaz InequalitySolve - ako príkaz na riešenie nerovníc

```
<< Algebra`InequalitySolve`
```

```
InequalitySolve[x / 3 - 1 / 2 > 1 / 6 + x, x]
```

$x < -1$

Mathematica 6 ho nahradila príkazom **Reduce.** Podľa mojich zistení sú v prípadoch stredoškolských nerovníc tieto príkazy ekvivalentné.

```
Reduce[x / 3 - 1 / 2 > 1 / 6 + x, x]
```

```
x < -1
```

Podobne to platí aj v ďalšom príklade

```
InequalitySolve[Abs[x - 2] + 3 ≤ 2 x, x]
```

$x \geq \frac{5}{3}$

```
Reduce[Abs[x - 2] + 3 ≤ 2 x, x]
```

$x \geq \frac{5}{3}$

Kvadratické rovnice - grafické riešenie

```
f[x_] = x^2 - 4 x - 5;
Table[{x, f[x]}, {x, -3, 6, 1}] // Transpose
```

```
{{-3, -2, -1, 0, 1, 2, 3, 4, 5, 6}, {16, 7, 0, -5, -8, -9, -8, -5, 0, 7}}
```

```
Table[{x, f[x]}, {x, -3, 6, 1}] // Transpose // TableForm
```

```
-3  -2  -1  0   1   2   3   4   5  6
16  7   0   -5  -8  -9  -8  -5  0  7
```

V Mathematice 6 už môžeme veľmi jednoducho vytvoriť aj tabuľku v slušnom tvare

```
Grid[Table[{x, f[x]}, {x, -3, 6, 1}] // Transpose,
 Dividers → All, ItemSize → {3, 1}]
```

-3	-2	-1	0	1	2	3	4	5	6
16	7	0	-5	-8	-9	-8	-5	0	7

Mathematica 5 umožňovala len túto formu preznetácie a to dosť komplikovane

```
f[x_] = x^2 - 4 x - 5;
Table[{x, f[x]}, {x, -3, 6, 1}] // Transpose;
TableForm[%, TableAlignments → Right]
```

−3	−2	−1	0	1	2	3	4	5	6
16	7	0	−5	−8	−9	−8	−5	0	7

```
Plot[f[x], {x, -3, 6}]
```



Vlastnosti kvadratickej funkcie - ako vplývajú parametre na tvar grafu

Najskôr si ukážeme formu prezentácie, ktorú umožňovala Mathematica 5

```
f[x_, c_] := x^2 - 4 x + c
```

```
Table[f[x, c], {c, -2, 4}]
```

$\{x^2 - 4x - 2, x^2 - 4x - 1, x^2 - 4x, x^2 - 4x + 1, x^2 - 4x + 2, x^2 - 4x + 3, x^2 - 4x + 4\}$

```
Plot[Evaluate[%], {x, -3, 6}]
```



- Graphics -

Mathematica 6 už môže použiť aj dynamickú vizualizáciu

```
Manipulate[Plot[x^2 - 4 x + c,
  {x, -3, 6}, PlotRange → {-20, 30}], {c, -2, 4}]
```



Mathematica 6 už môže použiť aj dynamickú vizualizáciu - môžeme súčasne modelovať zmenu viacerých parametrov

```
Manipulate[
 Plot[a x^2 + b x + c, {x, -5, 5}, PlotRange → {-30, 30}],
 {a, -3, 3}, {b, -3, 3}, {c, -10, 10}]
```



Rovnice s parametrom

```
Solve[Sqrt[x^2 + m] == m - x, x]
```

$$\left\{\left\{x \to \frac{1}{2}(-1 + m)\right\}\right\}$$

```
Clear[rovnica]
rovnica = Sqrt[x^2 + m] == m - x
```

$$\sqrt{m + x^2} == m - x$$

Mathematica 5 nedokáže urobiť automaticky kompletnú analýzu riešenia vzhľadom na daný parameter, treba najskôr urobiť analýzu diskriminantu

```
Solve[(m - 1) x^2 - (m - 2) x + 2 m - 1 == 0, x]
```

$$\left\{\left\{x\to\frac{-2+m-\sqrt{8\,m-7\,m^2}}{2\,(-1+m)}\right\},\left\{x\to\frac{-2+m+\sqrt{8\,m-7\,m^2}}{2\,(-1+m)}\right\}\right\}$$

```
diskriminant = Reduce[8 m - 7 m^2 >= 0, m]
```

$$0\le m\le\frac{8}{7}$$

```
Plot[8 m - 7 m^2, {m, -1, 2}]
```



Až po analýze diskriminantu sme dokázali pomocou príkazu inequality Solve analyzovať túto rovnicu. *Mathematica* 5 používala na riešenie nerovnice príkaz InequalitySolve

```
InequalitySolve[
 {(m - 1) x^2 - (m - 2) x + 2 m - 1 == 0, 0 ≤ m ≤ 8/7}, {m, x}]
```

$$m==0\,\&\&\,x==1\,||$$
$$0<m<1\,\&\&\left(x==\frac{-2+m}{2\,(-1+m)}-\frac{1}{2}\sqrt{\frac{8\,m-7\,m^2}{(-1+m)^2}}\;||\;x==\frac{-2+m}{2\,(-1+m)}+\frac{1}{2}\sqrt{\frac{8\,m-7\,m^2}{(-1+m)^2}}\right)||$$
$$m==1\,\&\&\,x==-1\;||\;1<m<\frac{8}{7}\,\&\&$$
$$\left(x==\frac{-2+m}{2\,(-1+m)}-\frac{1}{2}\sqrt{\frac{8\,m-7\,m^2}{(-1+m)^2}}\;||\;x==\frac{-2+m}{2\,(-1+m)}+\frac{1}{2}\sqrt{\frac{8\,m-7\,m^2}{(-1+m)^2}}\right)||\;m==\frac{8}{7}\,\&\&\,x==-3$$

Mathematic a 6 požíva príkaz Reduce. Vidíme, že výsledok je v oboch prípadoch rovnaký

```
Reduce[{(m - 1) x^2 - (m - 2) x + 2 m - 1 == 0, 0 ≤ m ≤ 8/7},
  {m, x}] // Simplify
```

$$(m == 0 \&\& x == 1) \,||\, (m == 1 \&\& 1 + x == 0) \,||\, \left(m == \frac{8}{7} \&\& 3 + x == 0\right) \,||\, \left(\left(0 < m < 1 \,||\, 1 < m < \frac{8}{7}\right) \&\& \left(\frac{m}{-1+m} + \sqrt{\frac{(8-7m)\,m}{(-1+m)^2}} == 2\left(\frac{1}{-1+m} + x\right) \,||\, \frac{2}{-1+m} + \sqrt{\frac{(8-7m)\,m}{(-1+m)^2}} + 2x == \frac{m}{-1+m}\right)\right)$$

Mathematic a 6 má ale vylepšený algoritmus na riešenie nerovníc a tak dokáže túto rovnicu s parametrom riešiť priamo, bez akýchkoľvek dodatočných podmienok.

```
Reduce[(m - 1) x^2 - (m - 2) x + 2 m - 1 == 0, x]
```

$$(m == 1 \&\& x == -1) \,||\, \left(-1 + m \neq 0 \&\& \left(x == \frac{-2 + m - \sqrt{8m - 7m^2}}{2(-1+m)} \,||\, x == \frac{-2 + m + \sqrt{8m - 7m^2}}{2(-1+m)}\right)\right)$$

Pripomeňme si ešte raz, ako vyzeral výsledok v Mathematice 5

```
Solve[(m - 1) x^2 - (m - 2) x + 2 m - 1 == 0, x]
```

$$\left\{\left\{x \to \frac{-2 + m - \sqrt{8m - 7m^2}}{2(-1+m)}\right\}, \left\{x \to \frac{-2 + m + \sqrt{8m - 7m^2}}{2(-1+m)}\right\}\right\}$$

Kreslíme funkcie - odstránenie problémov

Problémy, ktoré sme mali s kreslením funkcií tam, kde neboli vskutočnosti definované boli odstránené

```
Plot[Tan[x], {x, -2 Pi, 2 Pi}]
```




```
Plot[Tan[x], {x, -2 Pi, 2 Pi}, Exclusions → {1 / Tan[x] == 0}]
```




```
Plot[Log[x], {x, -3, 13}]
```



```
Plot[Sqrt[x], {x, -3, 13}, Exclusions → {Log[x] == 1}]
```



Grafické princípy riešenia lineárnych nerovníc

Budeme porovnávať riešenie rovnice a riešenie nerovnice $y = x+1$ a $y < x+1$

```
Plot[x + 1, {x, -2, 2}, Frame → True, AspectRatio → 1]
```



```
RegionPlot[y < x + 1, {x, -2, 2}, {y, -2, 2}, Axes → True]
```



A ako to bude s viacerými nerovnicami?

$y < 1/2x + 3$
$y > = -x - 1$

```
RegionPlot[y < 1 / 2 x + 3 && y ≥ -x - 1,
 {x, -5, 5}, {y, -5, 5},
 PlotPoints → 35, ColorFunction → "BlueGreenYellow"]
```



```
obr1 = RegionPlot[y < 1 / 2 x + 3 ,
  {x, -5, 5}, {y, -5, 5}, PlotPoints → 35]
```




```
obr2 = RegionPlot[ y ≥ -x - 1, {x, -5, 5}, {y, -5, 5},
  PlotPoints → 35, ColorFunction → "BlueGreenYellow",
  PlotStyle -> Opacity[0.5]]
```



```
Show[obr1, obr2, Axes → True]
```



Môžeme ilustrovať aj použitie niekoľkých logických operátorov. Na príklade obrázkov sa tieto súvislosti vysvetľujú veľmi jednoducho. Definujeme najskôr dve kružnice, rovnakého polomeru, len vzájomne posunuté. Následne vytvoríme tabuľku, ktorá bude prezentovať ich vzájomné vzťahy pomocou logických operátorov.

$$a = \left(-\frac{1}{2} + x\right)^2 + y^2 < 1;$$

$$b = \left(\frac{1}{2} + x\right)^2 + y^2 < 1;$$

```
Table[{RegionPlot [a, {x, -2, 2}, {y, -2, 2}],
  RegionPlot [a, {x, -2, 2}, {y, -2, 2}]}]
```



```
Table[RegionPlot[f[a, b], {x, -2, 2}, {y, -2, 2},
  PlotLabel → f], {f, {And, Or, Xor, Implies, Nand, Nor}}]
```



2. ročník – základné vlastnosti funkcií

Grafické možnosti systému môžeme prezentovať veľmi výhodne na vysvetlení základných vlastností funkcií

Ukážeme, ako parameter ovlyvňuje charakter funkcie

```
Manipulate[Plot[a Abs[x] + b, {x, -3, 3},
  PlotRange → {-10, 10}], {a, -5, 5}, {b, -5, 5}]
```



Párnosť, nepárnosť

```
f[x_] = Sqrt[0.2 x]
```

$0.447214 \sqrt{x}$

Na grafe funkcie vysvetlíme vlastnosti f(x) a f(−x)

```
Plot[{f[x], f[-x]}, {x, -15, 15}, PlotRange → {0, 3}]
```



Na grafe funkcie vysvetlíme vlastnosti f(x) a −f(−x)

```
Plot[{f[x], -f[-x]}, {x, -15, 15}, PlotRange → {-3, 3}]
```



3. ročník - analytická geometria

Vzájomná poloha troch rovín

Budeme vizualizovať nasledujúce tri roviny

$2x - y + z = 5$, $x + 3y + 2z = 4$, $x + 2y - 4z = -6$

```
Plot3D[{-2 x + y + 5, 1 / 2 x + 3 / 2 y - 4, 1 / 4 x + 2 / 4 y + 6},
 {x, -10, 10}, {y, -10, 10},
 Boxed → False, Mesh → False, Axes → False]
```



```
Plot3D[{-2 x + y + 5, 1 / 2 x + 3 / 2 y - 4, 1 / 4 x + 2 / 4 y + 6},
 {x, -10, 10}, {y, -10, 10}]
```



```
Manipulate[Plot3D[{-2 x + y + 5, 1 / 2 x + 3 / 2 y - c},
  {x, -10, 10}, {y, -10, 10}], {c, -10, 10}]
```



3. ročník - vlastnosti postupností a radov

```
Clear[a]
```

```
a[n_] := a[n] = 1 / 2 (3 a[n - 1] - a[n - 2]); a[1] = 2; a[2] = 1;
```

```
Table[a[n], {n, 15}]
```

$\left\{2, 1, \frac{1}{2}, \frac{1}{4}, \frac{1}{8}, \frac{1}{16}, \frac{1}{32}, \frac{1}{64}, \frac{1}{128}, \frac{1}{256}, \frac{1}{512}, \frac{1}{1024}, \frac{1}{2048}, \frac{1}{4096}, \frac{1}{8192}\right\}$

```
Clear[a, t, n]
```

```
a[n_] = (2 n - 1) / 5
```

$\frac{1}{5}(-1+2n)$

```
t = Table[a[n], {n, 15}]
```

$\left\{\frac{1}{5}, \frac{3}{5}, 1, \frac{7}{5}, \frac{9}{5}, \frac{11}{5}, \frac{13}{5}, 3, \frac{17}{5}, \frac{19}{5}, \frac{21}{5}, \frac{23}{5}, 5, \frac{27}{5}, \frac{29}{5}\right\}$

```
ListPlot[t, PlotStyle → {PointSize[0.02]}]
```




```
a[n + 1] - a[n] // Simplify
```

$\frac{2}{5}$

Step by step riešenie kvadratickej rovnice

Implementation

Usage message

```
WalkSolve::usage = "WalkSolve[q == 0, x] solves for the
two solutions of the quadratic equation q == 0 in the
unknown x one step at a time.";
```

Definitions

Explanation of the program

The variables e1, e2, e3, etc., are equations. We get the next equation by applying an algebraic operation to both sides of the previous equation. In general these equations will all be different, but it does happen in special cases that there is no change from one to the next. For example, if c = 0, e1 and e2 would be the same. WalkSolveSteps1 eliminates such repetitions in the lines redefining p up to taking the square root; WalkSolveSteps2 eliminates the rest. WalkSolve puts these two pieces together and deals with the case of a double root.

(Note that f @ g @ h @ x means f[g[h[x]]]. This notation saves a lot of nested brackets.)

```
WalkSolveSteps1[q_, a_, b_, c_] := Module[
    {e1, e2, e3, e4, e5, e6, p},
    e1 = q == 0;
    e2 = -c + First @ e1 == -c + Last @ e1;
    e3 = First @ e2 / a == Last @ e2 / a;
    e4 = Expand @ First @ e3 == Last @ e3;
    e5 = b ^ 2 / (4 a ^ 2) + First @ e4 ==
         b ^ 2 / (4 a ^ 2) + Last @ e4;
    e6 = Factor /@ e5;
    p = If[e1 === e2, {e1}, {e1, e2}];
    p = If[e2 === e3, p, Append[p, e3]];
    p = If[e3 === e4, p, Append[p, e4]];
    p = If[e4 === e5, p, Append[p, e5]];
    p = If[e5 === e6, p, Append[p, e6]];
    Do[Print[ p[[i]] ], {i, Length @ p}];
    e6
]
```

```
WalkSolveSteps2[x_, e6_, sign_] := Module[
    {e7, d, e8, e, e9, f, e10, p},
    e7 = PowerExpand @ Sqrt @ First @ e6 ==
    sign PowerExpand @ Sqrt @ Last  @ e6;
    d = Denominator @ First @ e7;
    e8 = d First @ e7 == d Last @ e7;
    e = First @ e8 /. x -> 0;
    e9 = First @ e8 - e == Last @ e8 - e;
    f = First @ e9 /. x-> 1;
    e10 = First @ e9 / f == Last @ e9 / f;
    p = If[e7 === e8, {e7}, {e7, e8}];
    p = If[e8 === e9, p, Append[p, e9]];
    p = If[e9 === e10, p, Append[p, e10]];
    Do[Print[ p[[i]] ], {i, Length @ p}];
]
```

```
WalkSolve[q_ == 0, x_] := Module[
    {a, b, c, e6},
    {a, b, c} = Reverse @ CoefficientList[q, x];
    e6 = WalkSolveSteps1[q, a, b, c];
If[
    Last @ e6 === 0,

(* then *)

    Print["Dostali sme koreň kvadratickej rovnice."];
    WalkSolveSteps2[x, e6, 1];
    Print["Toto je dvojnásobný koreň."],

(* else *)

    Print["Najskôr budeme uvažovať kladný koreň."];
    WalkSolveSteps2[x, e6, 1];
    Print["Dostali sme prvé riešenie."];
    Print["Teraz zoberieme druhý koreň."];
    WalkSolveSteps2[x, e6, -1];
    Print["Dostali sme druhé riešenie."];
]
]
```

```
rovnica = 2 - 4 x + x^2
```

$2 - 4x + x^2$

```
WalkSolve[rovnica == 0, x]
```

$2 - 4x + x^2 == 0$

$-4x + x^2 == -2$

$4 - 4x + x^2 == 2$

$(-2 + x)^2 == 2$

Najskôr budeme uvažovať kladný koreň.

$-2 + x == \sqrt{2}$

$x == 2 + \sqrt{2}$

Dostali sme prvé riešenie.

Teraz zoberieme druhý koreň.

$-2 + x == -\sqrt{2}$

$x == 2 - \sqrt{2}$

Dostali sme druhé riešenie.

```
rovnica2 = x ^ 2 + 2 B x + C;
```

```
WalkSolve[rovnica2 == 0, x]
```

$x^2 + 2Bx + C = 0$

$x^2 + 2Bx = -C$

$B^2 + 2xB + x^2 = B^2 - C$

$(B + x)^2 = B^2 - C$

Najskôr budeme uvažovať kladný koreň.

$B + x = \sqrt{B^2 - C}$

$x = \sqrt{B^2 - C} - B$

Dostali sme prvé riešenie.

Teraz zoberieme druhý koreň.

$B + x = -\sqrt{B^2 - C}$

$x = -B - \sqrt{B^2 - C}$

Dostali sme druhé riešenie.